
FORTRAN is an acronym that stands for "formula translation."  The

language was invented around 1957.  Historically, it was the first major

effort at a language that has a strong resemblance to English, a great

improvement over assembler languages.  Since it was the first such

effort, it was rather primitive in comparison with modern languages.

     ANSI standards published in 1977 made a large number of modernizing

improvements in FORTRAN.  One of the more highly used implementations of

FORTRAN-77 is VAX FORTRAN, which is presented in this chapter.  This form

of FORTRAN is available on VAX systems.  We also point out the

differences between VAX FORTRAN and FORTRAN-77, to help you read and

understand other implementations.  Subsections titled "Additional

Features of FORTRAN-77" briefly describe some features of FORTRAN-77 that

you might someday encounter, though not used in this book.

##1  REAL AND INTEGER TYPES

     Two numeric types, REAL and INTEGER, are discussed in this section.

REAL numeric values have decimal points, and INTEGER numeric values
do not.  In particular, 1.0 is a REAL value and 1 is an INTEGER value.

     FORTRAN is card-oriented.  In other words, rules about what

things you can write are stated in terms of their location on an

imaginary punch-card 72 characters wide.  The following material should

tell you enough to understand Example ##1A; the card-oriented rules are

in the first four parts.  Refer to Example ##1A frequently, line by line,

while studying these points.

  @  If a "C" or "*" is in column 1, the rest of that card is a comment,

     to be ignored by the compiler.  The purpose of a comment is to

     explain something about the program to someone who reads it.  No

     other form of comment is allowed in a FORTRAN program.  The rest of

     the card-oriented rules below apply only if the card is not a

     comment.

  @  Columns 1 through 5 are reserved for labels (often used to mark the

     target of a GOTO statement).  These are sequences of digits.  The

     sample program does not have any labels.

  @  If Column 6 contains any character, the compiler treats that line as

     if it were a continuation of the previous line.  This is because

     ends-of-lines are highly significant in FORTRAN.  The sample program

     uses only an underscore symbol to mark continuation lines, namely,

     for Lines 7, 15, and 16.

  @  All program material other than that described above must be in

     columns 7 through 72.   WARNING:  Anything you put beyond column

     72 is completely ignored.  So going outside these bounds often does

     not cause a compiler error.  Instead, the answers are mysteriously

     wrong.  The line numbers shown at the far right are ignored.

  @  Line 1:  You put the heading PROGRAM Calculate Taxes (or whatever

     the name you choose for the program) as the first line of the

     program.  The last line of the program must be END.

  @  Lines 2 & 3:  REAL Sales Tax Rate PARAMETER (Sales Tax Rate=0.075)

     declares that Sales Tax Rate is the name of the REAL constant 0.075.

     It is called a constant because the program is not allowed to

     change its value.

EXAMPLE ##1A _________________________________________________

      program    Calculate Taxes                                         1

C         Accept from the user at the keyboard a number of items

C         purchased and the amount paid per item; then report

C         the total cost including sales tax.

C     Written by Dr. William C. Jones  November 1993

* CONSTANTS

      real  Sales Tax Rate                                               2

      parameter (Sales Tax Rate = 0.075)                                 3

* VARIABLES

      integer Number Purchased                                           4

      real  Cost Per Item, Cost Without Tax, Sales Tax                   5

* BEGIN

* accept from the user the number of items and

*        the cost of each item

      print*,  'This is a little calculator program.',                   6

     _          '  How many items did you buy? '                         7

      read*, Number Purchased                                            8

      print*, 'How much did you pay for each one? '                      9

      read*, Cost Per Item                                              10

      print*                                                            11

* compute net cost and sales tax

      Cost Without Tax = Cost Per Item * REAL (Number Purchased)        12

      Sales Tax  = Cost Without Tax * Sales Tax Rate                    13

* report the total cost including sales tax

      print*,  'The total cost of ', Number Purchased,                  14

     _         ' items including tax is ',                              15

     _           Cost Without Tax + Sales Tax                           16

      end                                                               17

____________________________________________________________

  @  Line 4:  INTEGER Number Purchased declares that Number Purchased is

     used as the name of an INTEGER variable in this program.  A variable
     is a location in the computer's storage area (called its memory)

     where values can be stored.  So you can store one INTEGER numeric

     value in the memory location whose name is Number Purchased.  All

     declarations of variables must come before the statements.

  @  Line 5:  REAL... declares that the three names listed are used as

     the names of REAL variables in this program.  So you can store one

     REAL numeric value in each of these three storage locations.

  @  Line 6:  PRINT* is a command that causes the specified material to

     appear on the user's screen.  The material can be any number of

     REAL, INTEGER, or STRING expressions, preceded by commas.  This

     particular print* command prints two STRING expressions called

     string literals.  The beginning and end of a string literal are

     marked using an apostrophe.  The material between the apostrophes

     will appear on the screen exactly as shown, including spaces.  The

     underscore in column 6 below print* marks a line continuation.

  @  Line 8:  READ* is followed by any number of variables each preceded

     by a comma.  This causes that same number of values to be read from

     the user's keyboard and stored in the variables named, in the order

     listed.  The input values that the user types must be separated from

     each other by either commas or spaces.  The input values for one

     READ must be all on one line; the input values for two separate

     READs must be on separate lines.  So READ*,X,Y reads two numbers.

  @  Line 12:  Cost Without Tax = causes the value of the expression

     following the assignment symbol = to be calculated and then stored

     in the REAL variable named Cost Without Tax.  The * symbol is the

     multiplication operator.  The addition operator is the + symbol and

     the subtraction operator is the - symbol.

  @  Line 12:  REAL(Number Purchased) converts a copy of the INTEGER

     value stored in the location named Number Purchased to a REAL value

     so it can be multiplied by the REAL value in Cost Per Item.  This is

     called a type conversion from INTEGER to REAL.  However, this does

     not change the value that is still stored of Number Purchased.

     NOTE 1:  The use of the type conversion can be omitted -- FORTRAN

allows REALS and INTEGERS to be combined using operators and a value of

either numeric type to be assigned to a variable of the other type.

     NOTE 2:  The compiler ignores all spaces in a statement except those

within string literals.  For instance, "RE ALSales TaxRate" is the same

as "REAL Sales Tax Rate".  Most programming languages do not allow this.

     NOTE 3:  VAX FORTRAN allows up to 31 characters in an identifier,

which can be digits, uppercase, lowercase, or underscores.  So the name

Sales Tax could have been written as sales_tax throughout the program.

VAX FORTRAN is not case sensitive -- SALes and sALes and SALES are all

the same name.  By contrast, FORTRAN-77 only allows up to six characters,

none of which can be lowercase letters.

     NOTE 4:  All of the output from one PRINT statement appears on one

complete line.  Two PRINT statements always produce two lines, and so on.

     NOTE 5:  Within a string literal, two consecutive apostrophes are

used to represent a single apostrophe. So if you want to print "He's it",

you use the string literal 'He''s it'.

                                                                             4

SAMPLE RUN:  If this program is executed and the user enters 10 and

3.50 on separate lines, the following will appear on the screen:

   This is a little calculator program.  How many items did you buy?

   10

   How much did you pay for each one?

   3.50

   The total cost of 10 items including tax is 37.625

     The next material is a group of exercises.  You must work out the

answer to every exercise on a piece of paper.  Then you must compare your

answer to the correct one.  The correct answers will always be at the

bottom of the page after the exercises.  If you do not do this, you will

think you understand this material but you will find out you are wrong

when you take a test on it, but then it will be too late.

EXERCISE 1.1:  If the input for the execution of this program had been 3

and 2.00, what would have been printed on the last line of the output?

EXERCISE 1.2:  If the * symbol in Line 13 is changed to the + symbol,

how will that change the last line of the sample run shown at the top of

this page?

EXERCISE 1.3:  What change would you make in Line 9 of the program to

have it ask  "What's the cost of each?" instead?

EXERCISE 1.4:  What change would you make in the program so that it

prints 37.625 on a separate line after printing "including tax is"?

Real and Integer Expressions

     The REAL function used in Line 12 of Example ##1A converts one

INTEGER value to a REAL value (such as from 10 to 10.0) so it can be

multiplied by another REAL value.  To convert a REAL value to an INTEGER

value, use the INT function; it truncates (chops off the part

after the decimal point) whatever REAL value is in its parentheses.

     We recommend the use of the INT and REAL functions for clarity;

however, they are not always required.  FORTRAN allows you to mix INTEGER

and REAL values in an expression; evaluation of an operation combining an

INTEGER and a REAL value will cause conversion of the INTEGER value to a

REAL first.  Assignment of an INTEGER value to a REAL variable also

automatically converts the INTEGER value to a REAL value.  And assignment

of a REAL value to an INTEGER variable automatically applies the INT

function to that REAL value.  The following are some illustrations of

statements that could legally appear in Example ##1A:

C If Cost Per Item is 3.72, this puts 3 in Number Purchased

      number purchased = INT(Cost Per Item)

C But so does this

      number purchased = Cost Per Item

C If sales tax is 3.7, this then puts 6.7 (3.7 + 3.0) in sales tax

      sales tax = sales tax + number purchased

     There are five operators that can be used with REAL and INTEGER

values:  ** denotes exponentiation (as in 2 ** 5 = 32), * denotes

multiplication (as in 2 * 5 = 10), + denotes addition, - denotes

subtraction, and / denotes division.  / for INTEGER values gives the

quotient for integer division.  To get the remainder, use the MOD

function of two values.

     Some examples of the results of calculations with division are as

follows:

20 / 7 = 2, the quotient         MOD (20 , 7) = 6, the remainder

23 / 5 = 4, the quotient         MOD (23 , 5) = 3, the remainder

16 / 4 = 4, the quotient         MOD (16 , 4) = 0, the remainder

00 / 3 = 0, the quotient         MOD ( 0 , 3) = 0, the remainder

The divisor for / or MOD cannot be zero, since it does not make sense to

divide by zero.  Warning:  Failure to explicitly use the conversion

functions INT and REAL can lead to very subtle errors.  For instance,

X = 6.0 * 7 + 4 / 5 stores 42.0 in X, because 6.0*7 is 6.0*7.0 = 42.0,

and 4/5 is 0, so X is 42.0 + 0 which is 42.0 + 0.0 which is 42.0.  Note

that automatic type conversions are not made until necessary.

     Any variable that appears in an expression must have already had a

value assigned to it by execution of a previous statement.  Otherwise

the program may crash.  In particular, variables cannot be assumed to be

initially zero.  For instance, in Example ##1A, if the first statement

had been print*, Cost Per Item, before it was given a value, then you

would not know what would be printed -- 0.0 or 3172.53 or -4000000 or

whatever.

EXAMPLE ##1B _________________________________________________

      program    Illustrate Operators

C         Show how operators are used in FORTRAN

* BEGIN

C  This prints 4  3  0

      print*, 13 / 3,  15 / 5, 8 / 10

C  This prints 1  0  8

      print*, mod (13, 3), mod (15, 5), mod (8, 10)

      end

____________________________________________________________

EXERCISE 1.5:  What does this print:  \ print*, 3**4, 4**3, (-2)**3

EXERCISE 1.6:  What does this print:  \ print*, 23/4, 317/10, 37/12

EXERCISE 1.7:  What does this print:  \ print*, mod(23,4), mod(317,10)

EXERCISE 1.8:  A program is to read 3 integers into three variables named

A, B, and C.  What statements do you write if all three are on one line?

What statements do you write if all three are on three different lines?

ANSWERS TO EXERCISES:====================================================

For exercises, \ denotes the beginning of a FORTRAN statement

1.1:  The total cost of 3 items including tax is 6.45

1.2:  The total cost of 10 items including tax is 70.075

1.3:  Change it to \ print*, 'What''s the cost of each? '

1.4:  Take the comma off the end of Line 15 and change Line 16 to

      \ print*, Cost Without Tax + Sales Tax

=========================================================================

    FORTRAN does not require you to declare variables if you do not want

to.  The use of an undeclared variable beginning with any of the letters

I through N causes an implicit declaration as an INTEGER variable.  The

use of any other undeclared variable causes an implicit declaration as a

REAL variable.  This feature has the strong disadvantage that, if you

misspell an identifier, the compiler will not warn you as it does in most

modern languages.  Some people have spent hours searching for a bug in a

program when this was the only problem.  When you compile a program in

VAX FORTRAN, use FORTRAN/WARNINGS=DECLARATIONS as the command to have the

compiler warn you of a potential bug of this nature.

     GOOD PROGRAMMING STYLE:  You are expected to make your program very

easy to read.  So use a PROGRAM heading, declare all variables, and use

INT() and REAL() when converting numeric values from one type to another.

Also, do not capitalize all or even most of your program lines.  Any

variable name that is made up of two or more English words is to use one

of three techniques to make the separate words clear -- use all small

letters except (a) capitalize the first letter of each word (as in

SalesTaxRate) or (b) use underscores to separate the words (as in

sales_tax_rate) or (c) put spaces between the words (as in sales tax

rate) optionally with capitalizing the first letter of each word.

EBNF Descriptions

Some EBNF definitions (extended Backus-Naur formalism) for an elementary

FORTRAN program block (the part after the PROGRAM line) are as follows.

CR stands for the carriage-return (end-of-line).  Notation:  material

enclosed in the braces { } can occur 0 or more times, and vertical bars

separate several forms of the concept being defined.

     $ program-block = { declaration }  { statement }  END  CR

     $ declaration   = type-denoter  identifier  { , identifier }   CR

                     | PARAMETER ( const-decl { , const-decl } )    CR

     $ type-denoter  = REAL

                     | INTEGER

     $ const-decl    = identifier = constant-value

     $ statement     = READ *   { , identifier }  CR

                     | PRINT *  { , expression }  CR

                     | identifier = expression   CR

     The above definitions are complete except for an identifier (defined

as a letter followed by any number of letters, digits, and underscores),

a constant-value (a numeric value or a string literal), and an expression

(which is very messy to define).

     The type-denoters such as REAL and INTEGER are reserved words and

cannot be redefined as variable names.  By contrast, INT is a built-in

identifier and so should not be redefined as a variable name.

ANSWERS TO EXERCISES:====================================================

1.5:  81  64  -8

1.6:  5  31  3

1.7:  3  7

1.8:  if all on one line  \ read*, A, B, C

      if on 3 different lines:  read*, A  \  read*, B  \  read*, C

=========================================================================

